********************************************************************************************************************************
***   Replication file for:                                                                                                  ***
***   Berbee, P., Braun, S. T. and Franke, R. (2024). Reversing Fortunes of German Regions, 1926-2019. JoEG.			     ***
***   							                                                                                             ***
***   SCRIPT: 	_x2e_preparation_census_1950.do																				 ***	
***   PURPOSE: 	Reads in and prepares data from the 1950 occupation, population, and housing census						 	 ***
********************************************************************************************************************************

* Preamble (unnecessary when executing run.do)
run "$reversing/scripts/programs/_config.do"

************
* Code begins
************

*** Import data

*** Occupation census 

import excel "$reversing/data/Berufszaehlung_1950.xlsx", sheet("Data") first clear

destring T_ID, replace
rename T_ID labor_market_id

** Employment shares
foreach sector in wg3 wg4 wg5 wg8 wg9 wa0 wg11 wg12	wg13 wg14 wg15 wg17	wg19 wa1 wg21 wg22 wg23	wg24 wg25 wg26 wg27	wg28 wg29 wa2 wg31 wg32	wg34 wg35 wg36 wg37	wg38 wg39 wg41 wg42	wg43 wg44 wg45 wg46	wg47 wg48 wg49 wa3_4 wg53 wg54 wg55	wg56 wg57 wg59 wa5 wg65_6 wg67 wg68	wg69 wa6 wg71 wg72 wg73	wg74 wg75 wg76 wg77	wg78 wg79 wa7 wg81 wg82	wg83 wg84 wg85 wg86	wg87 wa8 wg91 wg92 wg93	wg94 wg95 wg96 wg97	wg98 wg99 wa99a	wa9	{
gen empshare_1950_`sector'=laborforce_1950_`sector'/laborforce_1950_total	
}

* Employment share in agriculture (including forestry & fishing)
gen empshare_1950_agriculture=laborforce_1950_wa0/laborforce_1950_total
gen laborforce_1950_agriculture=laborforce_1950_wa0

* Employment share in "Montanindustrie"
* Kohle- & Erzbergbau, Eisen-, Stahl-, Metallerzeugung sowie Stahl- und Waggonbau
gen laborforce_1950_montanindustries=(laborforce_1950_wg11+laborforce_1950_wg12+laborforce_1950_wg13+laborforce_1950_wg21+laborforce_1950_wg22+laborforce_1950_wg23)
gen empshare_1950_montanindustries=laborforce_1950_montanindustries/laborforce_1950_total

* Employment share in "modern industries"
* Maschinen-, Straßenfahrzeug-, Luftfahrzeugbau, Elektrotechnik, Feinmechanik, Optik, Chemische und Kunststoffindustrie
gen laborforce_1950_modernindustries=(laborforce_1950_wg24+laborforce_1950_wg26+laborforce_1950_wg27+laborforce_1950_wg28+laborforce_1950_wg31+laborforce_1950_wg32+laborforce_1950_wg34+laborforce_1950_wg35)
gen empshare_1950_modernindustries=laborforce_1950_modernindustries/laborforce_1950_total

* Employment in other industry and manufacturing
egen laborforce_1950_otherindustry=rowtotal(laborforce_1950_wg14 laborforce_1950_wg15 laborforce_1950_wg17 laborforce_1950_wg19 laborforce_1950_wg25 laborforce_1950_wg29 laborforce_1950_wg36 laborforce_1950_wg37 laborforce_1950_wg38 laborforce_1950_wg39 laborforce_1950_wg41 laborforce_1950_wg42 laborforce_1950_wg43 laborforce_1950_wg44 laborforce_1950_wg45 laborforce_1950_wg46 laborforce_1950_wg47 laborforce_1950_wg48 laborforce_1950_wg49)
gen empshare_1950_otherindustry=laborforce_1950_otherindustry/laborforce_1950_total

* Employment in construction
gen laborforce_1950_construction=laborforce_1950_wa5
gen empshare_1950_construction=laborforce_1950_construction/laborforce_1950_total

* Employment in the service sector
gen laborforce_1950_services=laborforce_1950_wa6+laborforce_1950_wa7+laborforce_1950_wa8+laborforce_1950_wa9
gen empshare_1950_services=laborforce_1950_services/laborforce_1950_total

* Employment share in all industries
gen empshare_1950_allind=(laborforce_1950_wa1+laborforce_1950_wa2+laborforce_1950_wa3_4)/laborforce_1950_total


global lmodernindustries modern industries
global lmontanindustries Montanindustrie
global lagriculture	agriculture
global lconstruction construction
global lservices services
global lotherindustry other industries
global lwg3		Landwirtschaft und Tierzucht
global lwg4		Forst- und Jagdwirtschaft
global lwg5		Gärtnerei
global lwg8		Hochsee- und Küstenfischerei
global lwg9		Binnenfischerei
global lwa0		Landwirtschaft(1digit)
global lwg11 	Steinkohlegewinnung und -verarbeitung
global lwg12 	Braunkohlegewinnung und -verarbeitung
global lwg13 	Erzbergbau
global lwg14 	Salzbergbau und Salinen
global lwg15 	Sonstiger Bergbau
global lwg17 	Abbau von Steinen und Erden, Baustoff- und Keramische Industrie
global lwg19 	Energiewirtschaft
global lwa1		Bergbau (1 digit)
global lwg21	Eisen- und Stahlerzeugung
global lwg22	NE-Metallerzeugung und -gießereien
global lwg23	Stahl- und Waggonbau
global lwg24	Machinen- und Apparatbau
global lwg25	Schiffbau
global lwg26	Straßenfahrzeig- und Luftfahrzeugbau
global lwg27	Elektrotechnik
global lwg28	Feinmechanik und Optik
global lwg29	Eisen-, Stahl- und Metallwarenfertigung
global lwa2		Eisen- und Metallerzeugung und -verarbeitung (1 digit)
global lwg31	Mineralölverarbeitung und Kohlewertstoffind.
global lwg32	Chemische Grundindustrie
global lwg34	Kunststoffverarbeitung
global lwg35	Gummi- und Asbestverarbeitung
global lwg36	Feinkeramische und Glasindustrie
global lwg37	Sägerei und Holzbearbeitung
global lwg38	Holzverarbeitung
global lwg39	Papiererzeugung und -verarbeitung, Druckereigew.
global lwg41	Ledererzeugung und -verarbeitung
global lwg42	Textilgewerbe
global lwg43	Bekleidungsgewerbe
global lwg44	Herst. von Musikinstrumenten, Spielwaren, Sportgeräten, Schmuckwaren
global lwg45	Mühlengewerbe, Nähr- und Futtermittelind., Bäckerei- und Süßwarengewerbe
global lwg46	Fleisch- und Fischverarbeitung, Milchverwertung, Speisefettherstellung, Zuckerindustrie
global lwg47	Obst- u. Gemüseverwertung, Kaffee-, Tee- u. Gewürzverarbeitung, Eisgewinnung und Frosterei
global lwg48	Getränkeherstellung
global lwg49	Tabakwarenherstellung
global lwa3_4	Verarbeitende Gewerbe (1 digit)
global lwg53	Architektur-, Bauingenieur- und Vermessungsbüros
global lwg54	Hoch- und Tiefbau
global lwg55	Zimerei und Dachdeckerei
global lwg56	Bauinstallation, Klempnerei und Elektroinstallation
global lwg57	Ausbaugewerbe
global lwg59	Bauhilfsgewerbe
global lwa5		Bau-, Ausbau- und Bauhilfsgewerbe (1 digit)
global lwg65_6	Warenhandel und Verlagsgewerbe
global lwg67	Wirtschaftswerbung und sonstige Hilfsgewerbe des Handels
global lwg68	Geld-, Bank- und Börsenwesen
global lwg69	Versicherungswesen
global lwa6		Handel, Geld- und Versicherungswesen (1 digit)
global lwg71	Wohnungs- und Grundstückswesen, Vermögensverwaltung
global lwg72	Gaststättenwesen
global lwg73	Kunst, Schrifttum, Theater, Film- und Rundfunkwesen, Schaust.-Gew.; priv. Forschung
global lwg74	Sportpflege, Bade- und Schwimmanstalten
global lwg75	Nachrichten-, Schreib- und Übersetzungsbüros
global lwg76	Photographisches Gewerbe
global lwg77	Friseurgewerbe 
global lwg78	Reinigungs-, Bewachsungs- und Trägergewerbe, priv. Bestattungswesen
global lwg79	Häusliche Dienste
global lwa7		Dienstleistungen (1 digit)
global lwg81	Deutsche Bundespost
global lwg82	Deutsche Bundesbahn
global lwg83	Schienenbahnen einschl. Obusbetriebe
global lwg84	Straßenverkehr
global lwg85	Schiffahrts-, Wasserstraßen- und Hafenwesen
global lwg86	Luftverkehr
global lwg87	Verkehrsneben- und -hilfsgewerbe
global lwa8		Verkehrswesen (1 digit)
global lwg91	Öffentliche Verwaltung
global lwg92	Besetzungsmächte und ausländische Vertretungen
global lwg93	Politische und wirtschaftliche Organisationen
global lwg94	Rechts- und Wirtschaftsberatung
global lwg95	Kirchen, weltanschauliche Vereinigungen
global lwg96	Erziehung, Wissenschaft, Kultur
global lwg97	Fürsorge und Wohlfahrtspflege
global lwg98	Sozialversicherung
global lwg99	Gesundheitswesen und Hygiene
global lwa9		Öffentlicher Dienst (1 digit)
global lwa99a	Ohne Angaben der Betriebszugehörigkeit

foreach sector in modernindustries montanindustries agriculture wg3	wg4	wg5	wg8	wg9	wa0	wg11 wg12 wg13 wg14	wg15 wg17 wg19 wa1 wg21	wg22 wg23 wg24 wg25	wg26 wg27 wg28 wg29	wa2	wg31 wg32 wg34 wg35	wg36 wg37 wg38 wg39	wg41 wg42 wg43 wg44	wg45 wg46 wg47 wg48	wg49 wa3_4 wg53	wg54 wg55 wg56 wg57	wg59 wa5 wg65_6	wg67 wg68 wg69 wa6 wg71	wg72 wg73 wg74 wg75	wg76 wg77 wg78 wg79	wa7	wg81 wg82 wg83 wg84	wg85 wg86 wg87 wa8 wg91	wg92 wg93 wg94 wg95	wg96 wg97 wg98 wg99 wa99a wa9 {
label var empshare_1950_`sector' "Employment share in ${l`sector'} (1950  occ.census)"
label var laborforce_1950_`sector' "Labor force in ${l`sector'} (1950  occ.census)"
}

preserve
keep labor_market_id laborforce_1950_*
tempfile laborforce_1950
save `laborforce_1950', replace
restore


foreach lab in modernindustries montanindustries agriculture construction services otherindustry{
label var laborforce_1950_`lab' "Labor force in ${l`lab'} (1950)"
label var empshare_1950_`lab' "Employment share in ${l`lab'} (1950)"
}
label var laborforce_1950_total "Total labor force (1950 occ.census)"
label var empshare_1950_allind "Employment share in industry (1950 occ.census)"

** Create HHI index for employment concentration in industry

* Global for 2-digit sectors in industry
global sectors2d1950 "wg11 wg12 wg13 wg14 wg15 wg17 wg19 wg21 wg22 wg23 wg24 wg25 wg26 wg27 wg28 wg29 wg31 wg32 wg34 wg35 wg36 wg37 wg38 wg39 wg41 wg42 wg43 wg44 wg45 wg46 wg47 wg48 wg49"

gen hhi_industryemp_1950 = 0
foreach sector in $sectors2d1950 {
gen hhi_summands_`sector' =  (laborforce_1950_`sector' / (laborforce_1950_wa1+laborforce_1950_wa2+laborforce_1950_wa3_4))^2
replace hhi_industryemp_1950 = hhi_industryemp_1950 + hhi_summands_`sector'
}

label var hhi_industryemp_1950 "HHI index of industry concentration, 1950"

*** Save data

keep labor_market_id empshare_1950_allind empshare_1950_modernindustries empshare_1950_montanindustries empshare_1950_otherindustry hhi_industryemp_1950

rename empshare_1950_allind empshare_ind_1950
rename empshare_1950_modernindustries empshare_modernindustries_1950 
rename empshare_1950_montanindustries empshare_montan_1950
rename empshare_1950_otherindustry empshare_otherindustry_1950


sort labor_market_id

save "$reversing/processed/intermediate/occ_census1950.dta",replace

*** Population census 1950 (expellee share)

** Import data

import excel "$reversing/data/Expellees_1950.xlsx", sheet("Data") first clear

rename Sheet1__La labor_market_id

label var pop_1950_hv "Total population 13/09/1950"
label var expellees_1950 "Total number of expellees 13/09/1950"

** Save data

keep labor_market_id pop_1950_hv expellees_1950
destring labor_market_id, replace
sort labor_market_id

save "$reversing/processed/intermediate/exp_census1950.dta",replace


*** Housing census 1950 (for war damages)

import excel "$reversing/data/Kriegsschaeden_1950.xlsx", sheet("Data") first clear

rename Sheet1__La labor_market_id

gen share_damaged_flats_1950 = flats_damaged_1950 / (flats_b1918_1950 + flats_1918_45_1950)

label var share_damaged_flats_1950 "Share of flats built 1945 or earlier that were damaged in the war"

** Save data

keep labor_market_id share_damaged_flats_1950
destring labor_market_id, replace
sort labor_market_id

save "$reversing/processed/intermediate/wardamage_census1950.dta",replace

*** EOF



